

Average Simulation of PWM Converters by Direct Implementation of Behavioral Relationships

Sam Ben-Yaakov*

Email: SBY@BGUEE.BITNET; Tel; +972-57-461561; FAX: +972-57-281340

Department of Electrical and Computer Engineering
Ben-Gurion University of the Negev
P. O. Box 653, Beer-Sheva, ISRAEL

Abstract - Average modeling of PWM converters is reexamined in the light of the behavioral dependent sources now included in modern versions of electronic circuit simulators. A topology independent behavioral model is developed to emulate the operation of PWM converters in voltage and current modes, for continuous and discontinuous inductor current cases. It is shown that, in general, the operation of the switching part involves three behavioral blocks: the Generic Switched Inductor Model (GSIM), the Duty Cycle Generator (DCG) and the Inductor Current Generator (ICG). Explicit expressions and equivalent circuits are developed for all possible modes of operation.

I. INTRODUCTION

The Switched Inductor Model (SIM) [1-5] developed for PWM and Quasi-Resonant converters is based on the average behavior of the switch mode *circuit*. This inherent feature and the fact that it is topology independent makes the model highly compatible with general purpose electronic circuit simulators.

The objective of this study was to reexamine the possibilities of direct average simulation of PWM converters in the light of the new developments in electronic circuit simulation technology.

II. MODERN ELECTRONIC CIRCUIT SIMULATORS

Most, if not all, modern electronic circuit simulators are based on SPICE [6] which was originally limited to polynomial dependent sources. This shortcoming of the earlier simulators has been alleviated in modern packages which now include generalized behavioral models. The behavioral dependent sources can be described by any algebraic relationship and hence the restriction of polynomial function has been removed. For example HSPICE (Meta-Software Inc., Campbell, CA) permits the definition of a behavioral dependent voltage source (EX) as:

$$EX = \frac{K V(a,b)}{ABS(V(c,d))}$$

where K is a constant, $V_{(m,n)}$ denotes the voltage between node (m) and (n) and ABS denotes absolute value (Please

note that the above form of the mathematical expression does not conform to the syntax of HSPICE).

Once defined, the behavioral relationship is fully compatible with other functions of the simulator such as .DC, .AC and .TRANS analyses.

III. THE BEHAVIORAL PWM AVERAGE MODEL

A prerequisite for a successful average simulation of PWM converters is the substitution of the switching parts with an average model. Close examination of the classical PWM topologies reveals that they all share a common, topology independent, switching module: a switched inductor [1-5]. Although normally realized by a switch and a steering diode, the switching action can be described by a toggle switch which alternately connects one end of the inductor to two terminals (b) and (c) (Fig.1)[1]. The interaction of the switched inductor module with the interface circuitry is described here by a Generic Switched Inductor Model (GSIM) which includes three dependent current sources (Fig. 2). The dependent sources of the GSIM represent the *average* terminal currents of the switched inductor assembly (Fig. 1). The behavioral description of these dependent current sources - i.e. their magnitude as a function of other parameters of the converter - will be a function of the operational mode of the converter.

The proposed behavioral PWM average model includes three basic blocks (Fig. 3): the GSIM, discussed above, the Inductor Current Generator (ICG) and the Duty Cycle Generator (DCG). Like the GSIM, the DCG function is algebraic. That is, it is the mathematical expression of the duty cycle in voltage mode and current mode converters. Since these functions are algebraic, the behavioral model of the DCG will be algebraic. In the case of the Inductor Current




Fig.1. The switched inductor assembly.

* Incumbent of the Luck-Hille Chair in Instrumentation Design




Fig 2. The Generic Switched Inductor Model (GSIM)

Generator (ICG), one has to distinguish four cases (Fig. 4). In the continuous inductor current cases, the inductor current is a dynamic function of the voltage across the GSIM terminals. As will be shown below, this differential equation relationship can conveniently be emulated by an equivalent circuit. For the discontinuous inductor current case, the relationship between the inductor current and the voltage across the GSIM terminals is algebraic and therefore can be described by an algebraic behavioral model.




Fig. 3. Behavioral model of the switching section of PWM converters.

	Continuous Inductor Current	Discontinuous Inductor Current
Voltage Mode	Dynamic	Algebraic
Current Mode	Dynamic	Algebraic

Fig. 4. The functional nature of the Inductor Current Generator (ICG) for all PWM cases.

IV. BEHAVIORAL MODULES OF PWM CONVERTERS

The algebraic and dynamic relationships for all basic PWM topologies were derived for all modes of operation and are summarized in APPENDIX I. The behavioral modules represent the average response of the modules as a function of the relevant parameters. The modules which involve a differential equation - i.e. the ICG in continuous mode - are represented by an equivalent circuit which is solved dynamically during simulation. Algebraic modules - e.g. the DCG - are represented by the corresponding mathematical relationships. These equations are applied by the simulator to calculate the relevant parameters during a large signal simulation. In the case of small signal (AC) simulation, the simulator will first linearize the equations around the 'DC' operating point before applying it in the (average) small signal simulation.

A. The ICG and the GSIM - continuous mode case.

The proposed topology-independent equivalent circuit is developed by considering the average voltage across the inductor. Examination of the switched inductor assembly (Fig. 1) reveals that the average voltage (V_L) is a function of the voltage across the terminals $\{V(a,b), V(a,c)\}$ and the time the switch is in the 'ON' (T_{ON}) and 'OFF' (T_{OFF}) positions:

$$V_L = \frac{T_{ON} V(a,b) + T_{OFF} V(a,c)}{T_S} \quad (1)$$

where T_S is the period of the switching cycle.

That is:

$$V_L = D_{on} V(a,b) + D_{off} V(a,c) \quad (2)$$

Since the duty ratios signals will be produced, in the present approach, by a dependent voltage sources (see below) we encode them into proportional voltage signals. The definition of the dependent voltage source (E_L) required to generate the average inductor voltage is thus:

$$E_L = V_{D_{on}} V(a,b) + V_{D_{off}} V(a,c) \quad (3)$$

Since the average inductor current in the continuous mode case is determined by the average voltage impressed on the inductor, the inductor current can be simulated in this case by the simple equivalent circuit which is given in APPENDIX I. The equivalent circuit includes the inductor, its parasitic resistance and an average voltage source (E_L) which, in general, will be time dependent. Also included in the equivalent circuit is a voltage source of zero voltage that is used to sense the current in the loop [6].

Once the inductor current is available, the dependent current sources of the GSIM model (Fig. 2) can be readily defined by considering the way the current is split between the terminals. Since the current of terminal (a) is identical to the inductor current we define the dependent current source (G_a) as:

$$G_a = I(V_{SN}) \quad (4)$$

where $I(V_{SN})$ is the current of the dummy voltage source (SN) of the equivalent circuit (APPENDIX I).

The current is then divided between terminals (a) and (b) (Fig. 1) according to the fraction of time the inductor is connected to each of them. Namely:

$$G_b = V_{D_{On}} I(V_{SN}) \quad (5)$$

$$G_c = V_{D_{Off}} I(V_{SN}) \quad (6)$$

Note that in the above equations the duty cycle is already encoded into voltage.

B. The ICG and the GSIM - discontinuous mode case.

Unlike the continuous inductor current case, the average inductor current in the discontinuous mode is an algebraic (and not a dynamic) function of the controlling parameters. This is a consequence of the fact that the peak inductor current (I_{pk}) is a linear function of the product ($V_{(a,b)} T_{ON}$) and that the terminal currents of the GSIM is related to the peak current by simple geometrical relationships (Fig. 5). The GSIM dependent current sources for the discontinuous case are hence defined as:

$$I_{G_a} = \frac{V_{D_{On}} + V_{D_{Off}}}{2} I_{pk} \quad (7)$$

$$I_{G_b} = \frac{V_{D_{On}}}{2} I_{pk} \quad (8)$$

$$I_{G_c} = \frac{V_{D_{Off}}}{2} I_{pk} \quad (9)$$

Where (Fig. 5):

$$I_{pk} = \frac{|V_{(a,b)}| V_{D_{On}}}{L} T_s \quad (10)$$




Fig. 5. The inductor current in the discontinuous case.

C. Duty Cycle Generator (DCG) - voltage mode cases .

For a given primary control voltage of the PWM modulator (V_E), the definition of the dependent voltage source ($E_{D_{On}}$) that generates the duty cycle signal ($V_{D_{On}}$) is:

$$E_{D_{On}} = K_M V_E \quad (11)$$

where K_M is the modulator's proportionality constant.

In the continuous mode case, D_{OFF} is related to D_{ON} by:

$$D_{OFF} = 1 - D_{ON} \quad (12)$$

Consequently:

$$E_{D_{Off}} = 1 - V_{D_{On}} \quad (13)$$

In the discontinuous mode case, the relationship between D_{OFF} and D_{ON} is :

$$\frac{|V_{(a,b)}| D_{ON}}{L} T_s = \frac{|V_{(a,c)}| D_{OFF}}{L} T_s = |I_{pk}| \quad (14)$$

where L is the inductance of the inductor.

Consequently:

$$E_{D_{Off}} = \frac{V_{D_{On}} |V_{(a,b)}|}{|V_{(a,c)}|} \quad (15)$$

D. Duty Cycle Generator (DCG) - current mode cases.

Applying the expression for the duty cycle generator for the continuous mode case [7], we define the duty cycle dependent voltage source as:

$$E_{D_{On}} = \frac{V_E - |I(V_{SN})| KS}{T_s (M_C + \frac{0.5 KS |V_{(a,b)}|}{L})} \quad (16)$$

where V_E is the primary control voltage (the output of the error amplifier), KS is the current loop gain and M_C is the slope of the compensating ramp.

In the current mode discontinuous inductor current case, the control voltage (V_E) is found to be (Fig. 6) :

$$V_E = (M_C + \frac{KS |V_{(a,b)}|}{L}) T_{ON} \quad (17)$$

Consequently, the duty cycle dependent voltage source is defined as:

$$E_{D_{On}} = \frac{V_E}{T_s (M_C + \frac{KS |V_{(a,b)}|}{L})} \quad (18)$$




Fig. 6. The inductor current in the current mode, discontinuous case.

V. APPLYING THE BEHAVIORAL MODEL

The topology independent behavioral modules developed above and summarized in APPENDIX I, can be used to develop input files for electronic circuit simulators which include algebraic behavioral models such as recent versions of HSPICE (Meta-Software Inc., Campbell, CA) and PSPICE (MicroSim Co., Irvine, CA). To simulate a given topology, one has to place the model in the corresponding orientation and in the discontinuous inductor current case, to set the

direction of the controlled current sources according to the expected directions of the currents. Other PWM topologies can be simulated by reducing them to one of the basic topologies. For example, the forward converter is treated as a Buck converter with a scaled input voltage, etc..

A PSPICE (MicroSim Co., Irvine, CA) compatible input file templates that covers all cases is given in APPENDIX II .

VI. EXAMPLE

The proposed simulation approach is demonstrated by considering a Current Mode (CM) buck converter (Fig. 7). The power stage is the benchmark circuit presented by Middlebrook [7] and investigated in an earlier study [3]. The average equivalent circuit (Fig. 8) was translated to a PSPICE compatible file by applying the template SUBCKT's given in APPENDIX II. Simulation results obtained by applying the proposed behavioral models were found to be in good agreement with previously published [4-6] analytical and experimental results.

The power of the proposed direct simulation approach is exemplified by the results of a number of key analyses. Fig. 9 presents the small signal control to output (v_{out}/v_E) transfer function for a 1.5Ω load. When the load is changed to 100Ω the converter's inductor current will be discontinuous and the equivalent circuit has to be switched to the discontinuous case per the relationships given in APPENDIX I. The corresponding control to output (v_{out}/v_E) transfer function for the discontinuous cases is given in Fig. 10.

Closed loop, large signal simulation is demonstrated by presenting the results of transient analysis when the converter is exposed to a load current change (Figs. 11, 12).

VII. DISCUSSION AND CONCLUSIONS

Simulation of the rather complex analog circuits presented here could be hampered by convergence problems. This is especially true in closed loop simulation in which the dynamic behavior of the system adds instability to the numerical algorithm used by the simulator. Nonetheless, it was found that in most of the cases the problem can be alleviated by following some simple procedural rules, which assist convergence. The followings are some suggestions that were found to be of great help in eliminating convergence deadlocks:

1. Evaluate the quiescent point voltages by running a 'DC' sweep in open loop configuration (V_{OUT} as a function of V_E).
2. Apply the operating point data obtained in step (1) to define initial voltage (by the .NODESET command) of critical nodes (e.g. output voltage, outputs of operational amplifiers etc.). It should be noted that the .NODESET command imposes a voltage source of relatively low input resistance. Consequently, it is advisable to add output resistance to amplifiers which are to be clamped

by the .NODESET instruction, assuming of course that it does not change the basic characteristics of the circuit.

3. Convergence problems in closed loop simulation are aggravated by fast non-linear devices and high gain error amplifiers. Therefore, do not use 'ideal' devices (diodes, transistors etc.) of zero or very low internal capacitances and reduce the (open loop) gain of the error amplifier as much as possible.
4. It should also be noted that the subcircuits given in APPENDIX II assume a specific mode of operation. They do not include a toggle arrangement to automatically switch from say, continuous inductor current mode to discontinuous mode [4]. Consequently one has to make sure that the converter is the assumed mode throughout a large signal simulation. This can be conveniently done by examining the sum:

$$\left\{ V_{D_{on}} + \frac{V_{D_{on}} |V_{(a,b)}|}{|V_{(a,c)}|} \right\} \quad (19)$$

which could be smaller than one for the discontinuous mode cases and equal to or larger than one for continuous mode cases. To help this evaluation, the SUBCKT's were designed with extra ports which carry the essential data (APPENDIX II).

The major advantage of the proposed topology independent models is the fact that they can be defined as subcircuit and directly applied to run DC, small signal (AC) and large signal (TRAN) analysis in open and closed loop configurations. By a simple modification [5] the model can be applied to simulate Average Current Mode systems [8] and to explore other schemes of single or multiple feedback configurations.

ACKNOWLEDGMENT

The simulation results presented in this paper were obtained by an evaluation version of PSPICE (V.5). The author wishes to thank MicroSim Co. for making this powerful demo public domain.




Fig. 7. Benchmark circuit [7] used to demonstrate the proposed simulation approach .




Fig. 8. PSPICE compatible equivalent circuit for the converter of Fig. 7 when operated in the continuous inductor current mode.




Fig. 9. Small signal control-to-output (v_{out}/v_E) transfer function of the power stage shown in Fig. 7. Continuous inductor current case (1.5Ω load).




Fig. 10. Small signal control-to-output (v_{out}/v_E) transfer function of the power stage shown in Fig. 7. Discontinuous inductor current case (100Ω load).




Fig. 11. Output voltage of the converter of Fig. 7 when subjected to a load change of 0.5Amp. Continuous inductor current case (1.5Ω load).




Fig. 12. Output voltage of the converter of Fig. 7 when subjected to a load change of 0.5Amp. Discontinuous inductor current case (100 Ω load).


REFERENCES

- [1] S. Ben-Yaakov, "SPICE simulation of PWM DC-DC converter systems: voltage feedback, continuous inductor conduction mode," *IEE Electronics Letters*, Vol. 25, No. 16, pp 1061-1063, August 1989.
- [2] S. Ben-Yaakov, D. Edry, Y. Amran and O. Shimony, "SPICE simulation of quasi-resonant zero-current-switching DC-DC converters," *IEE Electronics Letters*, Vol. 26, No. 13, pp 847-849, June 1990.
- [3] D. Kimhi and S. Ben-Yaakov, "A SPICE model for current mode PWM converters operating under continuous inductor current conditions," *IEEE Trans. on Power Electronics*, vol. 6, pp. 281-286, Apr. 1991.
- [4] Y. Amran, F. Huliehel and S. Ben-Yaakov, "A unified SPICE compatible average model of PWM converters," *IEEE Trans. on Power Electronics*, Vol. 6, pp. 585-594, Oct. 1991.
- [5] S. Ben-Yaakov and Z. Gaaton, "SPICE simulation of PWM DC-DC converter systems: voltage feedback, continuous inductor conduction mode," *IEE Electronics Letters*, Vol. 25, No. 16, pp 1061-1063, August 1989.
- [6] L.W. Nagel, "SPICE.2: A computer program to simulate semiconductor circuits", Memorandum No. ERL-M520, University of California, Berkeley, 1975.
- [7] R. D. Middlebrook, "Modeling current programmed Buck and Boost converters," *IEEE Trans. Power Electronics*, Vol. 4, pp. 36-52, January 1989.
- [8] L. H. Dixon, "Average current-mode control of switching power supplies", in *Unitrode switching regulated power supply design seminar manual* (Unitrode Co. Corporation, Merrimack, 1990), pp.5-1 - 5-14.

APPENDIX I

Behavioral Relationships and Equivalent Circuits of PWM Converters

AI.1 Generic Switched Inductor Model (GSIM).



AI.2 Definitions of Dependent Sources and Generators

	Continuous Inductor Current	Discontinuous Inductor Current
Inductor Current Generator (ICG)	$E_L = V_{Don} V_{(a,b)} + V_{Doff} V_{(a,c)}$	Algebraic
Generic Switched Inductor Model (GSIM) Sources	$G_a = I(V_{SN})$ $G_b = V_{Don} I(V_{SN})$ $G_c = V_{Doff} I(V_{SN})$	$ I_{pk} = \frac{ V_{(a,b)} }{L} V_{Don} T_s$ $ G_a = \frac{V_{Don} + V_{Doff}}{2} I_{pk} $ $ G_b = \frac{V_{Don}}{2} I_{pk} $ $ G_c = \frac{V_{Doff}}{2} I_{pk} $
Duty Cycle Generators (DCG) Voltage Mode	$E_{Don} = KM VE$ $E_{Doff} = 1 - V_{Don}$	$E_{Don} = KM VE$ $E_{Doff} = \frac{V_{Don} V_{(a,b)} }{ V_{(a,c)} }$
Duty Cycle Generators (DCG) Current Mode	$E_{Don} = \frac{VE - I(V_{SN}) KS}{T_s (MC + \frac{0.5 KS V_{(a,b)} }{L})}$ $E_{Doff} = 1 - V_{Don}$	$E_{Don} = \frac{VE}{T_s (MC + \frac{KS V_{(a,b)} }{L})}$ $E_{Doff} = \frac{V_{Don} V_{(a,b)} }{ V_{(a,c)} }$

AI.3 Notes

1. V_{Don} , V_{Doff} , I_{pk} , $V_{(a,b)}$, $V_{(a,c)}$, $I(V_{SN})$ are average, time dependent variables.
2. Don, Doff, are voltage coded (V_{Don} , V_{Doff} , respectively).
3. For discontinuous inductor current case, set G_a , G_b , G_c according to expected physical directions per the given topology.
4. The inputs to the model are VE (output signal of error amplifier) and the voltages across GSIM terminals (a), (b) and (c). All other variables are derived internally from the input signals and constants:
 - KS = Current gain
 - MC = Slope of compensating ramp [V/Sec]
 - KM = PWM modulation Constant [1/V]
 - L = Inductance [H]
 - T_s = Switching frequency period [Sec]

APPENDIX II

PSPICE (MicroSim Co.) TEMPLATES

PWM DEMO SUBCIRCUITS

* See Figs.7 & 8 of paper
 *Please NOTE: <expression> of 'VALUE' must be on one
 *line. Correct if shown otherwise in this printout.
 *Basic output circuit
 RO OUT 0 1.5; Load resistor
 CO C 0 2700U; Output capacitor
 RC OUT C .012 ; Capacitor's ESR
 *Input section
 VIN RS 0 30; 30 Volt source
 RS RS IN 0.105; Internal resistance of source
 *Error amplifier and phase compensation network
 *See Fig. 8 of main text for node labels
 ERR E 0 PLUS MINUS 1E4; Error Amplifier
 *VERR E 0 DC .5; For open loop simulation
 VREF PLUS 0 DC 7.5 AC 1 ; Reference plus perturbation
 RREF PLUS 0 1E5
 CF E EE 10n
 RF EE MINUS 10K
 R1 OUT MINUS 1K
 R2 MINUS 0 1K
 DDIOD 0 DIOD DN4X; Steering diode

*For DISCONTINUOUS cases
 *See SUBCKT "DISCONT" netlist for definition of ports
 *XSIM OUT IN DIOD E D DOF PK DISCONT

*For CONTINUOUS cases
 *See SUBCKT "CONT" netlist for definition of ports
 XSIM OUT IN DIOD E D DOF CONT

*Simulating a step in load current
 IOUT OUT 0 PULSE 0 500m 0.1m 10u 10u .5m

.TRAN 50U 1M
 .AC DEC 10 1 1MEG
 *.DC VERR 0 8 .5; For open loop simulation
 *.NODESET v(out)=15; To help convergence
 *Modified diode model
 .MODEL DN4X D(RS=.01 CJO=55PF IS=1.38E-09 N=1.7
 +VJ=.34V TT=5U M=.38 BV=2V)
 *Continuous conduction subcircuit
 .SUBCKT CONT A B C E D DOF
 * A B C => GSIM TERMINALS
 *E =>CURRENT MODE: OUTPUT OF ERRor AMP
 * VOLTAGE MODE: V(Don)
 * D => DUTY CYCLE INDICATOR (Don)
 *DOF => DUTY CYCLE INDICATOR (Doff)
 * Change according to parameters of circuit
 +PARAMS:LM =40U RL= .008 TS = 40U KS= 0.1 MC=5E4
 * INDUCTOR LM RES 1/Fs CUR GAIN SL COMP
 *Choose "EDD" for current mode, "ED" for voltage mode
 EDOF DOF 0 VALUE
 ={(V(E)*LM/ABS((V(A,C))*TS*KS+1U)); Doff indicator
 * Voltage mode

*ED DD 0 VALUE = {V(E)}
 * Current Mode
 EDD DD 0 VALUE ={(V(E)-
 ABS(I(VSN))*KS)/(TS*(MC+0.5*KS*ABS(V(A,B))/LM))}

*inductor Current Generator (ICG)
 VSN MI 0 0
 RL MR MI {RL}; Inductor's resistance
 LM M MR {LM}; inductor
 EM M 0 VALUE = { (V(A,B)*V(D)+V(A,C)*(1-V(D)))}
 *Generic Switched Inductor Model (GSIM)
 GOA A 0 VALUE= {I(VSN)}
 GOC 0 C VALUE = {I(VSN)*(1-V(D))}
 GOB 0 B VALUE = {I(VSN)*V(D)}

RVE E 0 10E10
 RD D 0 10E10
 RDOF DOF 0 1E10
 RDD DD D 10K
 DLM 0 D DN4X
 .ENDS

*Discontinuous conduction subcircuit
 .SUBCKT DISCONT A B C E D DOF PK
 * A B C => GSIM TERMINALS
 * E => VOLTAGE MODE: Don
 * D => DUTY CYCLE INDICATOR (Don)
 * DOF => DUTY CYCLE INDICATOR (Doff)
 * PK => INDUCTOR PEAK CURR INDICATOR

* Change according to parameters of circuit
 +PARAMS:LM =40U RL= .008 TS = 40U KS= 0.1 MC=5E4
 * INDUCTOR LM RES 1/Fs CUR GAIN SL COMP
 * Current Mode
 EDD DD 0 VALUE
 ={(V(E))/(TS*(MC+KS*ABS(V(A,B))/LM))}

* Voltage mode
 *ED DD 0 VALUE = {V(E)}

EPK PK 0 VALUE= {V(A,B)*V(D)*TS/LM}
 EDOF DOF 0 VALUE = {ABS(V(A,B)*V(D))/V(A,C)}
 *Generic Switched Inductor Model (GSIM)
 GA A 0
 VALUE={((V(A,B)*V(D)*TS/LM)*((V(D)+V(DOF))))/2}
 GC 0 C VALUE = {(V(A,B)*V(D)*TS/LM)*V(DOF)/2}
 GB 0 B VALUE = {(V(A,B)*V(D)*TS/LM)*V(D)/2}

RPK PK 0 1
 RVE E 0 10E10
 RD D 0 10E10
 RDOF DOF 0 1E10
 RDD DD D 10K
 DLM 0 D DN4X
 .ENDS
 .PROBE
 .END